
Relazione del programma di informatica: 
Editor di testi. 

 
Vincoli imposti dal problema : 
Progettare un editor che contenga una barra di menù con le seguenti voci: File, Modifica, 
Formato, ?. 
File ha le seguenti sottovoci: Nuovo, Apri, Salva, Salva con nome, Stampa, Esci. 
Modifica ha le seguenti sottovoci: Taglia, Copia, Incolla. 
Formato ha le seguenti sottovoci: Carattere, Colore. 
? ha le seguenti sottovoci: Giuda in linea, Informazioni su… 
Una barra dei controlli costituita da bottoni con relativa immagini dei comandi: Nuovo, Apri, 
Salva, Stampa, Taglia, Copia, Incolla, Grassetto, Corsivo, Sottolineato, Allinea a sinistra, 
Allinea al centro, Allinea a destra. 
Specifiche integrate :  
Per incollare anche immagini e testi preformatati, dopo la sottovoce “Incolla” ho aggiunto 
anche “Incolla speciale”. 
Per rendere più veloce e diretta la formattazione del testo ho aggiunto una seconda barra 
dei controlli con tre bottoni di stile “tbrPlaceholder” cioè, un pulsante che ha lo stesso 
aspetto e funzionamento di un separatore, ma è possibile impostarne la larghezza, ed al 
loro interno ho inserito due ComboBox e un’ImageCombo le loro funzioni sono: 

 ComboFont: permette di formattare e visualizzare il tipo di carattere del testo 
selezionato, contiene la lista di tutti i fonts visualizzabili sullo schermo. 

 ComboDim: consente la scelta la digitazione e la visualizzazione della dimensione 
dei caratteri selezionati; contiene ,per praticità, la lista dei numeri pari da 8 a 72 , 
ma è possibile inserire numeri interi compresi tra 1 e 2160. 

 ImageCombo1: permette di formattare e visualizzare il colore dei caratteri 
selezionati; contiene la lista dei 16 colori standard di Windows e in fine il colore 
RGB(xxx,xxx,xxx) che permette di aprire la finestra di dialogo per personalizzare i 
colori. Quando viene selezionato del testo con un colore standard viene visualizzato 
in colore e il suo nome, altrimenti un’immagine multicolore e il codice RGB. 

Gli altri elementi dell’interfaccia grafica sono un CommonDialog, due ImageList e una 
PictureBox le loro funzioni sono: 

 CommonDialog(DLGeditor): rende disponibile un gruppo di finestre di dialogo 
standard per l'esecuzione di operazioni quali l'apertura e il salvataggio di file, 
l'impostazione di opzioni di stampa e la selezione di colori e tipi di carattere. Il 
controllo consente inoltre di visualizzare la Guida in linea tramite l'esecuzione del 
motore della Guida di Windows. 

 ImageList(ImageList1): raccolta di immagini centrale che consente di fornire 
immagini ad altri controlli. L’ho utilizzata per contenere le immagini dei pulsanti della 
barra dei controlli. 

 ImageList(ImageList2): contiene i rettangoli con i 16 colori standard di windows e 
il 17° rettangolo multicolore. Le immagini sono disegnate in una PictureBox al 
caricamento del form, e vengono aggiunte con il metodo Add. 

 PictureBox (Picture1): impostando la proprietà BackColor si formano i rettangoli 
con i colori standard, mentre l’immagine multi colore e formata da 7 cerchi colorati. 

Facendo click con il tasto destro appare un menù a tendina con le opzioni: Taglia, Copia, 
Incolla, Carattere, Colore, Seleziona tutto. 
Questo menu permette all’utente di raggiungere velocemente le funzioni più usate. Per 
crearlo è stato necessario crearlo nell’editor dei menu con la proprietà Visible disattivata; 
con il metodo PopupMenu viene visualizzato nella posizione del puntatore. 

Legenda 

I  = input, O = output, L = lavoro, 

 

Struttura dati: 

Nome I O L Descrizione Tipo 

NomeFile    Percorso del File di editor String 

NomiColori    Vettore dei nomi di 16 colori standard Variant 

Indice    Indice della ImageCombo1 Byte 

R    Forma esadecimale del colore rosso String*4 

G    Forma esadecimale del colore verde String*4 

B    Forma esadecimale del colore blu String*4 

Queste variabili sono dichiarate nella sezione 
generale del form. 
 

 
Spiegazione del funzionamento del programma : 
Per comincia è possibile inserire un qualsiasi testo, il file aperto inizialmente è vuoto, può 
essere salvato clickando sulle voci del menù File o con il tasto CTRL+S. 
Per modificare il testo fare click con il tasto destro del mouse, oppure utilizzare i pulsanti 
della barra di controllo. Sono molto utili anche le caselle combinate per la scelta del tipo di 
carattere della dimensione dei caratteri e il colore. 

 
Codice   
Dim NomeFile As String 

Dim NomiColori As Variant 

Dim Indice As Byte 

Dim R, G, B As String * 4 

 
Private Sub ComboDim_Click() 

RTFtesto.SetFocus 

End Sub 

 
Private Sub ComboDim_KeyPress(KeyAscii As Integer) 

If KeyAscii = 13 Then RTFtesto.SetFocus 

End Sub 

 
Private Sub ComboDim_LostFocus() 

If Val(ComboDim.Text) < 1 Or Val(ComboDim.Text) > 2160 Or Not _ 

 IsNumeric(ComboDim.Text) Or ComboDim.Text = "  [Mista]" Then 

    MsgBox "Valore errato." & Chr(13) & "È possibile inserire solo numeri" _ 

                           & " interi compresi tra 1 e 2160!", vbCritical, _ 

                                         "Dimensione carattere non valida" 

    ComboDim.Text = "" 

    ComboDim.SetFocus 

    Exit Sub 

End If 

RTFtesto.SelFontSize = ComboDim.Text 

End Sub 

 
Private Sub ComboFont_Click() 

If ComboFont.Text = "[Tipo di carattere misto]" Then Exit Sub 

RTFtesto.SelFontName = ComboFont.Text 

RTFtesto.SetFocus 

End Sub 

 
Private Sub Form_Load() 

Dim i As Integer 

Dim Imagini As ComboItem 'Per inserire un immagine nell'elenco del ImageCombo1 

Dim Lista As ListImage 'Per inserire un immagine in ImageList2 

MnuPopup.Visible = False 

' Imposta la scala 

Picture1.Scale (9, 0)-(14, 2) 

' Popola la casella con i Fonts dello scermo. 

For i = 0 To Screen.FontCount - 1 

    ComboFont.AddItem Screen.Fonts(i) 

Next i 

FormEdit.Show 

ComboFont.Text = "Arial" ' Imposta su Arial. 

' Popola la casella con valori progressivi incrementati di 2 unità. 

For i = 8 To 72 Step 2 

   ComboDim.AddItem i 

Next i 

ComboDim.ListIndex = 1 ' Imposta dimensione 10. 

'Creazione della matrice con i nomi dei 16 colori. 

NomiColori = Array("Nero", "Blu", "Verde", "Azzurro", "Rosso", _ 

         "Fucsia", "Giallo", "Bianco", "Grigio", "Blu chiaro", _ 

             "Verde limone", "Azzurro chiaro", "Rosso chiaro", _ 

        "Fucsia chiaro", "Giallo chiaro", " Bianco brillante") 

' Popola la casella con le 16 imagini e i nomi dei  colori 

For i = 0 To 15 

    Picture1.BackColor = QBColor(i) 

    Set Lista = ImageList2.ListImages.Add(i + 1, NomiColori(i), Picture1.Image) 

    Set Imagini = ImageCombo1.ComboItems.Add(i + 1, NomiColori(i), _ 

                                    NomiColori(i), NomiColori(i)) 

Next i 

'Creazione ultima imagine multicolore 

Picture1.FillStyle = 0 

For i = 9 To 14 

    Picture1.FillColor = QBColor(i) 

    Picture1.Circle (i + 0.5, 1), 1, QBColor(i) 

Next i 

Set Lista = ImageList2.ListImages.Add(17, "RGB", Picture1.Image) 

Set Imagini = ImageCombo1.ComboItems.Add(17, "RGB", "RGB(xxx,xxx,xxx)", "RGB") 

ImageCombo1.ComboItems.Item(1).Selected = True 

Indice = 1 

End Sub 

 
Private Sub Form_Resize() 

'Ridimensionamento RichTextBox 

RTFtesto.Move 0, ToolBar1.Height, FormEdit.Width - 100, Abs(Toolbar2.Top - _ 

                                                           ToolBar1.Height) 

End Sub 

 
Private Sub ImageCombo1_Click() 

Indice = ImageCombo1.SelectedItem.Index 

RTFtesto.SetFocus 

End Sub 

 
Private Sub ImageCombo1_KeyPress(KeyAscii As Integer) 

If KeyAscii = 13 Then RTFtesto.SetFocus 

End Sub 

 
Private Sub ImageCombo1_LostFocus() 

On Error GoTo Errore 

ImageCombo1.ComboItems.Item(Indice).Selected = True 

If Indice = 17 Then 

    DLGeditor.Color = RTFtesto.SelColor 

    DLGeditor.Flags = cdlCCFullOpen Or cdlCCRGBInit 

    DLGeditor.ShowColor 

    RTFtesto.SelColor = DLGeditor.Color 

    R = "&H" & Right(Hex(DLGeditor.Color), 2) 

    G = "&H" & Mid(Hex(DLGeditor.Color), 3, 2) 

    B = "&H" & Left(Hex(DLGeditor.Color), 2) 

    ImageCombo1.ComboItems.Item(17).Text = "RGB(" & CStr(Val(R)) & "," _ 

                             & CStr(Val(G)) & "," & CStr(Val(B)) & ")" 

Else 

    RTFtesto.SelColor = QBColor(Indice - 1) 

End If 

Exit Sub 

'Gestisce tutti gli errori generati 

Errore: 

'Se l'utente ha scelto il pulsante Annulla. 

If Err.Number = 32755 Then Exit Sub 

Resume Next 'Se l'errore è un'altro 

End Sub 

 
Private Sub mnuEdit_Click() 

mnuEdit1(0).Enabled = (RTFtesto.SelLength > 0) 

mnuEdit1(1).Enabled = (RTFtesto.SelLength > 0) 

'Disativo taglia e copia 

End Sub 

 
Private Sub mnuEdit1_Click(Index As Integer) 

Select Case Index 

    Case 0 

        'Taglia 

        Clipboard.Clear 

        Clipboard.SetText RTFtesto.SelText 

        RTFtesto.SelText = "" 

    Case 1 

        'Copia 

        Clipboard.Clear 

        Clipboard.SetText RTFtesto.SelText 

    Case 2 

        'Incolla 

        RTFtesto.SelText = Clipboard.GetText() 

    Case 3 

        'Incolla speciale 

        RTFtesto.SelText = Clipboard.GetText(vbCFRTF) 

End Select 

End Sub 

 
Private Sub mnuFile1_Click(Index As Integer) 

On Error GoTo Errore 

Select Case Index 

    Case 0 

        'Nuovo 

        RTFtesto.Text = "" 

        FormEdit.Caption = "Editor - File nuovo" 

        NomeFile = "" 

    Case 1 

        'Apri 

        DLGeditor.ShowOpen 

        NomeFile = DLGeditor.FileName 

        RTFtesto.LoadFile NomeFile 

        FormEdit.Caption = "Editor - " & DLGeditor.FileTitle 

    Case 2 

        'Salva 

        If Len(NomeFile) = 0 Then 

            DLGeditor.ShowSave 

            NomeFile = DLGeditor.FileName 

        End If 

        RTFtesto.SaveFile NomeFile 

        FormEdit.Caption = "Editor - " & DLGeditor.FileTitle 

    Case 3 

        'Salva con nome... 

        DLGeditor.ShowSave 

        NomeFile = DLGeditor.FileName 

        RTFtesto.SaveFile NomeFile 

        FormEdit.Caption = "Editor - " & DLGeditor.FileTitle 

    Case 5 

        'Stanpa 

        DLGeditor.Flags = cdlPDReturnDC 

        DLGeditor.ShowPrinter 

        RTFtesto.SelPrint DLGeditor.hDC 

    Case 7 

        End 

End Select 

Exit Sub 

'Gestisce tutti gli errori generati 

Errore: 

'Se l'utente ha scelto il pulsante Annulla. 

If Err.Number = 32755 Then Exit Sub 

Resume Next 'Se l'errore è un'altro 

End Sub 

 
Private Sub mnuFormato1_Click(Index As Integer) 

On Error GoTo Errore 

Select Case Index 

    Case 0 

        'Carattere 

        With DLGeditor 

            .FontName = RTFtesto.SelFontName 

            .FontSize = RTFtesto.SelFontSize 

            .FontBold = RTFtesto.SelBold 

            .FontItalic = RTFtesto.SelItalic 

            .FontStrikethru = RTFtesto.SelStrikeThru 

            .FontUnderline = RTFtesto.SelUnderline 

            .Color = RTFtesto.SelColor 

        End With 

        DLGeditor.CancelError = True 

        DLGeditor.Flags = cdlCFBoth Or cdlCFEffects 

        DLGeditor.ShowFont 

        With RTFtesto 

            .SelFontName = DLGeditor.FontName 

            .SelFontSize = DLGeditor.FontSize 

            .SelBold = DLGeditor.FontBold 

            .SelItalic = DLGeditor.FontItalic 

            .SelStrikeThru = DLGeditor.FontStrikethru 

            .SelUnderline = DLGeditor.FontUnderline 

            .SelColor = DLGeditor.Color 

        End With 

    Case 1 

        'Colore 

        DLGeditor.Color = RTFtesto.SelColor 

        DLGeditor.Flags = cdlCCFullOpen Or cdlCCRGBInit 

        DLGeditor.ShowColor 

        RTFtesto.SelColor = DLGeditor.Color 

End Select 

Exit Sub 

'Gestisce tutti gli errori generati 

Errore: 

'Se l'utente ha scelto il pulsante Annulla. 

If Err.Number = 32755 Then Exit Sub 

Resume Next 'Se l'errore è un'altro 

End Sub 

 
Private Sub mnuHelp1_Click(Index As Integer) 

Select Case Index 

    Case 0 

        'Guida in linea 

        DLGeditor.HelpFile = "HelpFile" 

        DLGeditor.HelpCommand = cdlHelpContents 

        DLGeditor.ShowHelp 

    Case 1 

        'Informazioni Su... 

        MsgBox "Editor® 2006" & Chr(13) & "Prodotto da Cibelli-eNgeniar™" & _ 

         Chr(13) & "Copyright © 2000-2006. Tutti i diritti sono riservati." _ 

                & Chr(13) & "Programmato da Marco Cibelli.", vbInformation, _ 

                                                  "Informazioni Su Editor®" 

End Select 

End Sub 

 
Private Sub MnuPopup_Click() 

PopupMenu1(0).Enabled = (RTFtesto.SelLength > 0) 

PopupMenu1(1).Enabled = (RTFtesto.SelLength > 0) 

PopupMenu1(7).Enabled = (RTFtesto.SelLength <> Len(RTFtesto.Text)) 

'Disativo taglia, copia e Seleziona tutto; se non egeguibili. 

End Sub 

 
Private Sub PopupMenu1_Click(Index As Integer) 

Select Case Index 

    Case 0 

        'Taglia 

        mnuEdit1_Click (0) 

    Case 1 

        'Copia 

        mnuEdit1_Click (1) 

    Case 2 

        'Incolla 

        mnuEdit1_Click (2) 

    Case 4 

        'Carattere 

        mnuFormato1_Click (0) 

    Case 5 

        'Colore 

        mnuFormato1_Click (1) 

    Case 7 

        'Seleziona tutto 

        RTFtesto.SelStart = 0 

        RTFtesto.SelLength = Len(RTFtesto.Text) 

End Select 

End Sub 

 
Private Sub RTFtesto_MouseDown(Button As Integer, Shift As Integer, x As Single, 

y As Single) 

If Button = 2 Then 

   PopupMenu MnuPopup 

End If 

End Sub 

 
Private Sub RTFtesto_SelChange() 

On Error Resume Next 

' SelFontName restituisce il tipo di carattere o Null se è di tipo misto. 

If Not IsNull(RTFtesto.SelFontName) Then 

   ComboFont.Text = RTFtesto.SelFontName 

   If ComboFont.List(ComboFont.ListCount - 1) = "[Tipo di carattere misto]" Then 

        ComboFont.RemoveItem (ComboFont.ListCount - 1) 

   End If 

Else 

   ComboFont.AddItem "[Tipo di carattere misto]" 

   ComboFont.Text = "[Tipo di carattere misto]" 

End If 

' SelFontSize restituisce le dimensioni del carattere 

' o Null se i caratteri sono di dimensioni miste. 

If IsNull(RTFtesto.SelFontSize) Then 

    ComboDim.Text = "  [Mista]" 

Else 

    ComboDim.Text = RTFtesto.SelFontSize 

End If 

' SelColor restituisce il colore del carattere o Null se è di tipo misto. 

If IsNull(RTFtesto.SelColor) Then 

    ImageCombo1.ComboItems.Item(17).Text = "[Colore carattere misto]" 

    ImageCombo1.ComboItems.Item(17).Selected = True 

Else 

    ImageCombo1.ComboItems.Item(17).Text = "RGB(xxx,xxx,xxx)" 

    For Indice = 0 To 15 

        If RTFtesto.SelColor = QBColor(Indice) Then Exit For 

    Next Indice 

    Indice = Indice + 1 

    If Indice = 17 Then 

        'Ricava e stampa il colore RGB dal formato esadecimale. 

        R = "&H" & Right(Hex(RTFtesto.SelColor), 2) 

        G = "&H" & Mid(Hex(RTFtesto.SelColor), 3, 2) 

        B = "&H" & Left(Hex(RTFtesto.SelColor), 2) 

        ImageCombo1.ComboItems.Item(17).Text = "RGB(" & CStr(Val(R)) & "," _ 

                                 & CStr(Val(G)) & "," & CStr(Val(B)) & ")" 

    End If 

    ImageCombo1.ComboItems.Item(Indice).Selected = True 

End If 

'La proprietà value accetta 1 e 0 come impostazioni mentre SelBold, SelItalic, 

'SelUnderline restituiscono un espressione di tipo booleano (True o False) 

'rispettivamente -1 e 0. Quindi Abs(numero) restituisce il valore assoluto di 

'un numero è la sua grandezza senza segno. Ad esempio, ABS(-1) e ABS(1) 

'restituiscono entrambe 1. Cosi il -1 del True imposta value = 1. 

ToolBar1.Buttons(11).Value = Abs(RTFtesto.SelBold) 

ToolBar1.Buttons(12).Value = Abs(RTFtesto.SelItalic) 

ToolBar1.Buttons(13).Value = Abs(RTFtesto.SelUnderline) 

Select Case RTFtesto.SelAlignment 

    Case 0 

        'Allineamento a sinistra 

        ToolBar1.Buttons(15).Value = tbrPressed 

    Case 1 

        'Allineamento a destra 

        ToolBar1.Buttons(17).Value = tbrPressed 

    Case 2 

        'Allineamento al centro 

        ToolBar1.Buttons(16).Value = tbrPressed 

    Case Else 

        ToolBar1.Buttons(15).Value = tbrUnpressed 

        ToolBar1.Buttons(17).Value = tbrUnpressed 

        ToolBar1.Buttons(16).Value = tbrUnpressed 

End Select 

End Sub 

 
Private Sub ToolBar1_ButtonClick(ByVal Button As MSComctlLib.Button) 

Select Case Button.Key 

    Case "Nuovo" 

        mnuFile1_Click (0) 

    Case "Apri" 

         mnuFile1_Click (1) 

    Case "Salva" 

         mnuFile1_Click (2) 

    Case "Stanpa" 

         mnuFile1_Click (5) 

    Case "Taglia" 

         mnuEdit1_Click (0) 

    Case "Copia" 

         mnuEdit1_Click (1) 

    Case "Incolla" 

         mnuEdit1_Click (2) 

    Case "Grassetto" 

         RTFtesto.SelBold = Not RTFtesto.SelBold 

    Case "Corsivo" 

         RTFtesto.SelItalic = Not RTFtesto.SelItalic 

    Case "Sottolineato" 

         RTFtesto.SelUnderline = Not RTFtesto.SelUnderline 

    Case "Allineamento a sinistra" 

         RTFtesto.SelAlignment = 0 

    Case "Allineamento al centro" 

         RTFtesto.SelAlignment = 2 

    Case "Allineamento a destra" 

         RTFtesto.SelAlignment = 1 

End Select 

End Sub 

Interfacia: 

Ridimensionamento: 

 
 
 
 
 

 Firma________________ 


